2

Points, lines and circles

2.1 Equations of a Circle



The implicit equation of a circle

$$(x - x_J)^2 + (y - y_J)^2 - r_J^2 = 0$$

is the most commonly used for whole circles. The parametric form

$$x = x_J + r_J \cos\theta$$

$$y = y_J + r_J \sin\theta$$

is also straightforward, giving a parameterisation in terms of the angle subtended at the circle centre:



This form is particularly useful when the circle is the path of a rotating object, as the parameter, θ, can describe any number of rotations.

Alternatively, the parameterisation in terms of the half-angle subtended at the centre:

$$x = x_J + r_J \frac{(1 - t^2)}{(1 + t^2)}$$

$$y = y_J + r_J \frac{2t}{(1 + t^2)}$$

$$t = \tan \frac{\theta}{2}$$

eliminates the need for trigonometric functions. It defines the circle as follows:



but is normally only used to define a single quadrant, $0 < t < 1$, to avoid the obvious numerical problems.

Both parametric forms are useful for describing arcs; Section 3.4 gives details of this.

2.2 Intersections of a Line and a Circle



This problem is most easily solved if the circle is in implicit form

$$(x - x_J)^2 + (y - y_J)^2 - r_J^2 = 0$$

and the line is parametric:

$$x = x_0 + ft$$

$$y = y_0 + gt$$

Substituting the parametric equations into the circle equation gives a quadratic in t, the two roots of which represent the points on the line where it cuts the circle. If the roots are imaginary, then the line does not cut the circle at all. If the roots are coincident the line is tangential to the circle.



The value of t at the intersection points is

$$t = \frac{f(x_J - x_0) + g(y_J - y_0) \pm \sqrt{\{r_J^2(f^2 + g^2) - [f(y_0 - y_J) - g(x_0 - x_J)]^2\}}}{(f^2 + g^2)}$$

and the points are found by substituting these values back into the parametric equations. This is coded:

```
FSQ = F*F
GSQ = G*G
FGSQ = FSQ + GSQ
IF (FGSQ.LT.ACCY) THEN

        ..... Line coefficients are corrupt

ELSE
        XJO = XJ - XO
        YJO = YJ - YO
        FYGX = F*YJO - G*XJO
        ROOT = RJ*RJ*FGSQ - FYGX*FYGX
        IF (ROOT.LT.-ACCY) THEN

                ..... Line does not intersect circle

        ELSE
                FXGY = F*XJO + G*YJO
                IF (ROOT.LT.ACCY) THEN
                        T = FXGY/FGSQ          Line is tangential
                        X = XO + F*T
                        Y = YO + G*T
                ELSE
                        ROOT = SQRT(ROOT)    Two intersections
                        FGINV = 1.0/FGSQ
                        T1 = (FXGY - ROOT)*FGINV
                        T2 = (FXGY + ROOT)*FGINV
                        X1 = XO + F*T1
                        Y1 = YO + G*T1
                        X2 = XO + F*T2
                        Y2 = YO + G*T2
                ENDIF
        ENDIF
ENDIF
```

Note that, if the parametric line is normalised, the variable FGSQ will be 1.0, and the code can be simplified. To convert an implicit line equation to parametric form use the method described in Section 1.2

2.3 Intersections of Two Circles



Two circles may have two intersection points or one intersection point at a common tangent.



Alternatively they may not intersect at all. The position of the intersection points may be found by applying Pythagoras' theorem, which will give the parametric equation of the line on which the intersection points lie, and then solving the resulting quadratic equation in the parameter as was done in the last section. The substitution back into the parametric line equation can be done at the same time to shorten the code, which mirrors the algebra:

```
RKSQ = RK*RK          It is more efficient to use squared radii
RLSQ = RL*RL          and to omit this
XLK = XL - XK
YLK = YL - YK

DISTSQ = XLK*XLK + YLK*YLK
```

```
IF (DISTSQ.LT.ACCY) THEN

        ..... The two circles have the same centre

ELSE
        DELRSQ = RLSQ - RKSQ
        SUMRSQ = RKSQ + RLSQ
        ROOT = 2.0*SUMRSQ*DISTSQ-DISTSQ*DISTSQ-DELRSQ*DELRSQ
        IF (ROOT.LT.-ACCY) THEN

                ..... The circles do not intersect

        ELSE
                DSTINV = 0.5/DISTSQ
                SCL = 0.5 - DELRSQ*DSTINV
                X = XLK*SCL + XK
                Y = YLK*SCL + YK
                IF (ROOT.LT.ACCY) THEN

                        ..... Circles just touch at (X, Y)

                ELSE
                        ROOT = DSTINV*SQRT(ROOT) Two
                        XFAC = XLK*ROOT          intersections
                        YFAC = YLK*ROOT
                        X1 = X - YFAC
                        Y1 = Y + XFAC
                        X2 = X + YFAC
                        Y2 = Y - XFAC
                ENDIF
        ENDIF
ENDIF
```

The straight line between the points of intersection and, in the limit, the common tangent is given by the implicit equation

$$ax + by + c = 0$$

where

$$a = x_L - x_K$$

$$b = y_L - y_K$$

and $$c = \frac{[(r_L^2 - r_K^2) - (x_L - x_K)^2 - (y_L - y_K)^2]}{2} - x_K(x_L - x_K) - y_K(y_L - y_K)$$

This equation is not normalised.

2.4 Tangents from a Point to a Circle



If the point is outside the circle there are two tangents to it, if it is just on the circumference there is one, and if it is inside there is none.

If the equation of the tangent line required is

$$ax + by + c = 0$$

Then the coefficients a and b are obtained from:

$$a = \frac{\mp r_K(x_K - x_J) - (y_K - y_J)\sqrt{[(x_K - x_J)^2 + (y_K - y_J)^2 - r_K^2]}}{(x_K - x_J)^2 + (y_K - y_J)^2}$$

$$b = \frac{\mp r_K(y_K - y_J) + (x_K - x_J)\sqrt{[(x_K - x_J)^2 + (y_K - y_J)^2 - r_K^2]}}{(x_K - x_J)^2 + (y_K - y_J)^2}$$

and c can then be calculated from the fact that the tangent passes through J:

$c = -ax_J - by_J$

There are normally two possible tangent lines, obtained by attaching a sign to r_K. If the square root is positive then the two tangents are obtained as follows:



If the point J lies on the circle the contents of the square root are zero, and there is only a single tangent. If J is inside the circle there are no tangents, of course, and the root goes negative. Repeated factors make the corresponding code short:

```
XKJ = XK - XJ
YKJ = YK - YJ

XKJSQ = XKJ*XKJ
YKJSQ = YKJ*YKJ
DENOM = XKJSQ + YKJSQ
IF (DENOM.LT.ACCY) THEN

        ..... J and K are coincident

ELSE
        ROOT = DENOM - RK*RK
        IF (ROOT.LT.-ACCY) THEN

                ..... J is within the circle

        ELSE
                DENINV = 1.0/DENOM
                IF (ROOT.LT.ACCY) THEN
                        A = -RK*XKJ*DENINV        J lies on circle
                        B = -RK*YKJ*DENINV
                ELSE
```

```
                ROOT = SQRT(ROOT)
                RKSIGN = RK              Negate RKSIGN
                                         for other
                                         tangent

                A = (-YKJ*ROOT - RKSIGN*XKJ)*DENINV
                B = (XKJ*ROOT - RKSIGN*YKJ)*DENINV
            ENDIF
            C = -(A*XJ + B*YJ)
        ENDIF
    ENDIF
```

If the coordinates of the tangent point are required, they can be obtained from the a and b coefficients of the appropriate line:

$$x = x_K + ar_K$$

$$y = y_K + br_K$$

2.5 Tangents to a Circle Normal to a Line

This problem always has two solution lines, one each side of the circle.



If the known line is

$$a_1x + b_1y + c_1 = 0$$

and the circle is

$$(x - x_J)^2 + (y - y_J)^2 - r_J^2 = 0$$

then the two new lines tangential to the circle and normal to the given line will be

$$a_{2,3}x + b_{2,3}y + c_{2,3} = 0$$

where

$$a_{2,3} = \mp \frac{b_1}{\sqrt{(a_1^2 + b_1^2)}}$$

$$b_{2,3} = \pm \frac{a_1}{\sqrt{(a_1^2 + b_1^2)}}$$

and $c_{2,3} = r_J - a_{2,3}x_J - b_{2,3}y_J$

which can be coded:

```
ROOT = A1*A1 + B1*B1
IF (ROOT.LT.ACCY) THEN

        ..... The line equation is corrupt

ELSE
        DENOM = 1.0/SQRT(ROOT)
        AFAC = B1*DENOM
        BFAC = A1*DENOM
        A2 = AFAC
        B2 = -BFAC
        C2 = RJ - A2*XJ - B2*YJ
        A3 = -AFAC
        B3 = BFAC
        C3 = RJ - A3*XJ - B3*YJ
ENDIF
```

If the vector (a,b) points towards J from the given line, then lines 2 and 3 are as labelled in the diagram. Otherwise they are reversed. The tangent points are given by:

$$x_{2,3} = x_J + a_{2,3}r_J$$

$$y_{2,3} = y_J + b_{2,3}r_J$$

Note that the lines produced will be normalised, and that if the given line is normalised the code simplifies because ROOT and DENOM both become 1.0.

2.6 Tangents between Two Circles



If the tangent between the circles is

$$ax + by + c = 0$$

then its coefficients are given by:

$$a = \frac{(\mp r_L \pm r_K)(x_L - x_K) - (y_L - y_K)\sqrt{[(x_L - x_K)^2 + (y_L - y_K)^2 - (\pm r_L \mp r_K)^2]}}{(x_L - x_K)^2 + (y_L - y_K)^2}$$

$$b = \frac{(\mp r_L \pm r_K)(y_L - y_K) + (x_L - x_K)\sqrt{[(x_L - x_K)^2 + (y_L - y_K)^2 - (\pm r_L \mp r_K)^2]}}{(x_L - x_K)^2 + (y_L - y_K)^2}$$

$$c = \mp r_K - a x_K - b y_K$$

The signs attached to the two circle radii determine which of the four possible tangents



is to be found.

If the two circles intersect, only two tangents are possible:



Attempting to calculate either of the non-existant tangents will lead to a negative expression to be square rooted in the formula.

Finding a tangent may be coded:

```
RLK = RL - RK
```

Assumes both circles to be touched anticlockwise. RK and/or RL must be negated otherwise

```
YLK = YL - YK
XLK = XL - XK

XLKSQ = XLK*XLK
YLKSQ = YLK*YLK
```

```
DENOM = XLKSQ + YLKSQ
IF (DENOM.LT.ACCY) THEN

        ..... Circle centres are coincident

ELSE
        ROOT = DENOM - RLK*RLK
        IF (ROOT.LT.-ACCY) THEN

                ..... This tangent does not exist

        ELSE
                ROOT = SQRT(AMAX1(0.0,ROOT))
                DENINV = 1.0/DENOM
                A = (-RLK*XLK - YLK*ROOT)*DENINV
                B = (-RLK*YLK + YLK*ROOT)*DENINV
                C = -(RK + A*XK + B*YK)
        ENDIF
ENDIF
```

The line equation produced will be normalised.

2.7 Circles of Given Radius through Two Points



This is actually the same problem as finding the intersections of two circles with the same radius centred on the given points. These two intersection points will be the centres.



The same code can be used as that given in Section 2.3, with the added simplification that the value of the variable DELRSQ must be zero.

2.8 Circles of Given Radius through a Point and Tangent to a Line



This problem is really a special case of the one dealt with later in Section 2.11. The point J is first taken as a local origin. The line equation is referred to this origin by transforming the constant term, c:

$$c' = c + ax_J + by_J$$

The value of c' must be made positive, by multiplying the whole transformed equation by -1 if necessary. The circle centre coordinates are then found from the expressions

$$x = \frac{-a(c' - r_I) \pm b\sqrt{[r_I^2(a^2 + b^2) - (c'-r_I)^2]}}{a^2 + b^2}$$

$$y = \frac{-b(c' - r_l) \mp a\sqrt{[r_l^2(a^2 + b^2) - (c'-r_l)^2]}}{a^2 + b^2}$$

which simplify to

$$x = -a(c' - r_l) \pm b\sqrt{[c'(2r_l - c')]}$$

$$y = -b(c' - r_l) \mp a\sqrt{[c'(2r_l - c')]}$$

if the line is normalised. The two signs from the root give the two cases on one side of the line:



The root corresponding to the top signs (+ for x, − for y) gives the centre (x_1, y_1) to the left of the perpendicular from J to the line. The other root generates the centre (x_2, y_2) to the right of the perpendicular. A zero root indicates that there is only one possible circle, and a negative value indicates that the point is too far from the line for a circle to be created with the radius given. Lastly, note that if J is on the line, only one centre will be generated on the side of the line to which the vector (a,b) is pointing. This case is detected in the following code, *which assumes that the line is normalised.* For details of how to normalise a line equation see Section 1.2.

```
CDASH = C + A*XJ + B*YJ
IF (ABS(CDASH).LT.ACCY) THEN

        ..... Point J lies on the line.

ELSE
        IF (CDASH.LT.0.0) THEN
                ATEMP = -A
                BTEMP = -B
                CDASH = -CDASH
        ELSE
```

```
            ATEMP = A
            BTEMP = B
      ENDIF
      CFAC = CDASH - RI
      ROOT = RI*RI - CFAC*CFAC
      IF (ROOT.LT.-ACCY) THEN

            ..... Point J is too far from the line.

      ELSE
            IF (ROOT.LT.ACCY) THEN
                  X = XJ - ATEMP*CFAC          One possible
                  Y = YJ - BTEMP*CFAC          circle
            ELSE
                  ROOT = SQRT(ROOT)            Two possible
                  XCONST = XJ - ATEMP*CFAC     circles
                  YCONST = YJ - BTEMP*CFAC
                  XVAR = BTEMP*ROOT
                  YVAR = ATEMP*ROOT
                  X1 = XCONST + XVAR
                  Y1 = YCONST - YVAR
                  X2 = XCONST - XVAR
                  Y2 = YCONST + YVAR
            ENDIF
      ENDIF
ENDIF
```

2.9 Circles of Given Radius Tangent to Two Lines



lines is the circle diameter, an infinite number of solutions) then there are four centres for the circle of given radius that make it tangential to both the lines. These centres are distributed symmetrically about the point where the lines intersect.



If the two lines are

$$a_1x + b_1y + c_1 = 0$$

and $$a_2x + b_2y + c_2 = 0$$

and the given radius is r_l, then the circle centres are at:

$$x = \frac{b_2[c_1 \pm r_l\sqrt{(a_2^2 + b_2^2)}] - b_1[c_2 \pm r_l\sqrt{(a_1^2 + b_1^2)}]}{(a_2b_1 - a_1b_2)}$$

$$y = \frac{a_2[c_1 \pm r_l\sqrt{(a_1^2 + b_1^2)}] - a_1[c_2 \pm r_l\sqrt{(a_2^2 + b_2^2)}]}{(a_1b_2 - a_2b_1)}$$

Note that the denominator for y is minus the denominator for x. If the first r_l in each equation is made negative, the centre will be on the side of the first line corresponding to the vector (a_1, b_1); if positive it will be on the opposite side. The second r_l terms affect the centre position similarly with respect to the second line.

This is coded:

```
DETERM = A2*B1 - A1*B2
IF (ABS(DETERM).LT.ACCY) THEN

          ..... The lines are parallel

ELSE
          AB1 = SQRT(A1*A1 + B1*B1)
          AB2 = SQRT(A2*A2 + B2*B2)

          C1RAB1 = C1 + RI*AB1          To get the four solutions
          C2RAB2 = C2 + RI*AB2          change the + here to -

          DETINV = 1.0/DETERM
          X = (B2*C1RAB1 - B1*C2RAB2)*DETINV
          Y = (A1*C2RAB2 - A2*C1RAB1)*DETINV
ENDIF
```

2.10 Circles of Given Radius through a Point and Tangent to a Circle



One of the given points (circle centre, K, or J) is made a local origin for the calculation; we have chosen the point J. The circle centre coordinates may then be found from

$$x = x_J + \frac{x_{KJ}[(x_{KJ}^2 + y_{KJ}^2) - r_K(2r_I + r_K)] \pm y_{KJ}s}{2(x_{KJ}^2 + y_{KJ}^2)}$$

$$y = y_J + \frac{y_{KJ}[(x_{KJ}^2 + y_{KJ}^2) - r_K(2r_I + r_K)] \mp x_{KJ}s}{2(x_{KJ}^2 + y_{KJ}^2)}$$

where

$$x_{KJ} = x_K - x_J$$

$$y_{KJ} = y_K - y_J$$

and $$s = 4r_I^2(x_{KJ}^2 + y_{KJ}^2) - [(x_{KJ}^2 + y_{KJ}^2) - r_K(2r_I + r_K)]^2$$

The sign of r_K is important. If it is positive the two circles are outside each other, as in the diagram above. If r_K is negative, one is inside the other.



or



Positive and negative roots correspond to the two possible cases



Imaginary roots indicate that the circle is required to be outside the given circle and the given point is inside, or vice versa. A zero denominator indicates that the given point and the given circle centre are coincident. The coding is economical because of the many repeated sub-expressions.

```
XKJ = XK - XJ
YKJ = YK - YJ

SQSUM = XKJ*XKJ + YKJ*YKJ
IF (SQSUM.LT.ACCY) THEN
```

```
        ..... J and K are coincident

ELSE
        SQINV = 0.5/SQSUM
        RADSUM = (RI + RI + RK)*RK
        SUBEXP = SQSUM - RADSUM
        ROOT = 4.0*RI*RI*SQSUM - SUBEXP*SUBEXP
        SUBEXP = SUBEXP*SQINV
        IF (ROOT.LT.-ACCY) THEN

                ..... No centre possible

        ELSE
                IF (ROOT.LT.ACCY) THEN
                        X = XJ + XKJ*SUBEXP     Only one circle
                        Y = YJ + YKJ*SUBEXP     possible
                ELSE
                        ROOT = SQRT(ROOT)*SQINV
                        XCONST = XJ + XKJ*SUBEXP
                        YCONST = YJ + YKJ*SUBEXP
                        XVAR = YKJ*ROOT
                        YVAR = XKJ*ROOT
                        X1 = XCONST - XVAR
                        Y1 = YCONST + YVAR
                        X2 = XCONST + XVAR
                        Y2 = YCONST - YVAR
                ENDIF
        ENDIF
ENDIF
```

2.11 Circles of Given Radius Tangent to a Line and a Circle



This problem is approached, like that in Section 2.8, by taking the point J as a local origin. The line equation is referred to this origin by transforming the constant term, c:

$$c' = c + ax_J + by_J$$

The signs of the terms in this new equation are important. If the fixed circle crosses the line then the sign of c in the line equation determines whether solutions are found on the same or on the opposite side of the line as the fixed circle centre. The sign of r_J determines whether internal or external tangents are returned:



The equation should be multiplied through by -1 to give the sign of c required.

The circle centre coordinates are found from the expressions:

$$x = x_J + \frac{a(c' - r_I) \pm b\sqrt{[(a^2 + b^2)(r_I \pm r_J)^2 - (c' - r_I)^2]}}{(a^2 + b^2)}$$

$$y = y_J + \frac{b(c' - r_I) \mp a\sqrt{[(a^2 + b^2)(r_I \pm r_J)^2 - (c' - r_I)^2]}}{(a^2 + b^2)}$$

which simplify to

$$x = x_J + a(c' - r_I) \pm b\sqrt{[(r_I + r_J)^2 - (c' - r_I)^2]}$$

$$y = y_J + b(c' - r_I) \mp a\sqrt{[(r_I + r_J)^2 - (c' - r_I)^2]}$$

if the line was normalised. The two signs from the root give the two cases on one side of the line.



A zero root indicates that only one tangent circle is possible in the specified region. A negative number to be square rooted indicates that J is too far from the line for any tangents to be possible on the side of the line being considered of the type (internal/external) being sought.

Thus, *with the original line normalised* (see Section 1.2), we have code of the form:

```
CDASH = C + A*XJ + B*YJ
IF (CDASH.LT.0.0) THEN              Assumes tangent circles on J side
                                    are required, otherwise use .GE.

        ATEMP = -A
        BTEMP = -B
        CDASH = -CDASH
ELSE
        ATEMP = A
        BTEMP = B
ENDIF
CFAC = CDASH + RI
RFAC = RI + RJ                      Assumes external tangents required
                                    otherwise use RI - RJ

ROOT = RFAC*RFAC - CFAC*CFAC
IF (ROOT.LT.-ACCY) THEN

        ..... There are no solutions in this region

ELSE
        IF (ROOT.LT.ACCY) THEN
                X = XJ + ATEMP*CFAC          Only one circle possible
                Y = YJ + BTEMP*CFAC
        ELSE
                ROOT = SQRT(ROOT)            Two solutions
                XCONST = XJ - ATEMP*CFAC
                YCONST = YJ - BTEMP*CFAC
```

```
                X2 = XCONST + XVAR
                Y2 = YCONST - YVAR
            ENDIF
        ENDIF
```

2.12 Circles of Given Radius Tangent to Two Circles



This problem reduces to that given in Section 2.10 by transforming the given radii to make one of the fixed circles into a point with zero radius:

$$r_I' = r_I + r_J$$

$$r_J' = 0$$

$$r_K' = r_K - r_J$$

If this is done the centre of the new circle of radius r_I' will be in the position required for the centre of r_I. If the original radii are signed correctly (positive for external tangency, negative for internal) changes of sign deriving from the transformation will not affect the result.